


Application Note		資料作成 081127	資料番号 an2k-003
テーマ	RS-485 接続温湿度計の記録と取り出し その2。 資料番号 an2k-002 と似ていますが、このサンプルでは時刻と温湿度を MPC の点データエリアに記録します。		
使用機器	MPC-2100, MPC-Cunet, MRS-MCOM, USB-CUnet タッチパネル:パナソニック電工 GT11, 温湿度計:CHNO HN-CHNS Windows PC (VB6 オリジナルアプリ)		

■機器構成



■概要

- ・ 2 台の温湿度計のデータを一定間隔で MPC の点データエリアに蓄積保存します。
- ・ 必要に応じて PC を接続し、VB アプリケーションでデータを読み込みます。
- ・ VB アプリは CSV ファイル保存が出来るので、別途 EXCEL で処理をすることも可能です。
- ・ 通常は MPC が単独で稼働、任意のタイミングでデータを取得するという想定です。

※RS-485 結線、通信フォーマット、タッチパネル画面は資料番号 an2k-002 をご覧下さい。

■このサンプルの特徴

MPC は CU_POST サーバを起動し、メール機能でデータ転送をしています。
この場合、VB アプリがホスト、MPC がスレーブとなり、VB の要求に対して MPC がバックグラウンドで自動的に応答するので、MPC のユーザープログラムではメール機能を実行するコードはありません。

■MPC プログラム

```

CUNET 4 4 8 /* CUnet 初期化
CU_POST /* CUnet サーバ起動
MEWNET 38400 2 /* タッチパネル通信
S_MBK &H20 8 /* タッチパネルページ切り替え
OFF 70000 /* START SW オフ
CNFG# 5 RS485 "9600b7pes1NONE" /* MRS-MCOM ch5 RS-485 モード

FOR I=100 TO 131 /* タッチパネル使用エリア初期化
  S_MBK 0 I
NEXT I
S_MBK 0 200~Wrd /* Write Count

*START
S_MBK 0 300 /* タッチパネル Msg "Push START"
WAIT SW(70000)=1 /* タッチパネル START SW (オルタネート)
S_MBK 1 300 /* タッチパネル Msg "Logging"
DELAY=75 /* read 間隔 必要に応じて加減
NEWP /* 点データクリア
PNT=1 /* 点番号
OUT 0 2032~Int /* データ数@CUnet グローバルメモリ

FORMAT "0000/00/00" /* 文字列書式
S_MBK HEX$(DATE(0)) 124 10 /* タッチパネル 年月日表示

DO
  CH$="01" /* 温湿度計 ch1
  GOSUB *GET_DATA
  RES1$=RES$ /* タッチパネル 温度,湿度表示
  S_MBK RES1$ 100 9
  TEMP1=t
  HUM1=h
  TIME DELAY /* この温湿度計には必要みたい

  CH$="02" /* 温湿度計 ch2
  GOSUB *GET_DATA
  RES2$=RES$ /* タッチパネル 温度,湿度表示
  S_MBK RES2$ 108 9
  TEMP2=t
  HUM2=h

  TM=TIME(0) /* 時計
  FORMAT "00:00:00"
  S_MBK HEX$(TM) 116 8 /* タッチパネル 時分秒表示
  IF TM&&HFF==0 THEN /* 秒のチェック
    X(PNT)=TM /* 時間(HEX で入る)
    Y(PNT)=TEMP1*&H10000+HUM1 /* 上位 2byte=温度、下位 2byte=湿度
    Z(PNT)=TEMP2*&H10000+HUM2 /* 上位 2byte=温度、下位 2byte=湿度
    PR PNT P(PNT)


    WAIT (TIME(0)&&HFF) !=0 /* 秒!=00 待ち
    OUT PNT 2032~Int /* CUnet グローバルメモリに書き込み
    S_MBK PNT 200~Wrd /* タッチパネル カウンタ値表示
    PNT=PNT+1
  END_IF

  TIME DELAY /* この温湿度計には必要みたい
  IF SW(70000)==0 THEN /* タッチパネル STOP SW 押下なら
    GOTO *START
  END_IF
LOOP

*GET_DATA
PRINT# 5 CHR$(5) CH$ CHR$(2) "RPV01" CHR$(3) "¥r¥n" /* 温湿度計にコマンド送信
INPUT# 5 a$ /* 温湿度計データ受信
dummy=VAL(a$)
FOR i=1 to 9
  dummy=VAL(0)
NEXT I
t=VAL(10) /* 温度値取得
dummy=VAL(0) /* 読み飛ばし
h=VAL(10) /* 湿度値取得
FORMAT "00.0" /* 書式=整数 2 桁、小数 1 桁
RES$=STR$(t) /* 温度文字列
FORMAT "00.0"
RES$=RES$+", "+STR$(h) /* 温度文字列,湿度文字列
RETURN

```

■PCのVB6プログラム



◆Module1

usbuncnet.dll コマンド宣言 (サンプルで使っているのはこの内の一部です)

```

Declare Function usb_open Lib "usbuncnet.dll" () As Long
Declare Sub usb_close Lib "usbuncnet.dll" ()
Declare Function fw_vid Lib "usbuncnet.dll" () As Long
Declare Function fw_pid Lib "usbuncnet.dll" () As Long
Declare Function dll_ver Lib "usbuncnet.dll" () As Long
Declare Function fw_ver Lib "usbuncnet.dll" () As Long
Declare Sub init_cunet Lib "usbuncnet.dll" (ByVal sa As Long, ByVal ow As Long, ByVal en As Long)
Declare Function cunet_in Lib "usbuncnet.dll" (ByVal adr As Long, ByVal siz As Long) As Long
Declare Sub cunet_out Lib "usbuncnet.dll" (ByVal dat As Long, ByVal adr As Long, ByVal siz As Long)
Declare Sub cunet_on Lib "usbuncnet.dll" (ByVal adr As Long)
Declare Sub cunet_off Lib "usbuncnet.dll" (ByVal adr As Long)
Declare Function cunet_sw Lib "usbuncnet.dll" (ByVal adr As Long) As Long
Declare Function cunet_peek Lib "usbuncnet.dll" (ByVal adr As Long) As Long
Declare Sub cunet_poke Lib "usbuncnet.dll" (ByVal dat As Long, ByVal adr As Long)

Declare Function cunet_post_pnt Lib "usbuncnet.dll" (ByVal dest_sa As Long, ByVal ar_top As Long, ByRef send_ar As Any)
As Long
Declare Function cunet_req_pnt Lib "usbuncnet.dll" (ByVal req_sa As Long, ByVal ar_top As Long, ByRef rcv_ar As Any)
As Long

Public Declare Sub Sleep Lib "kernel32" (ByVal dwMilliseconds As Long)

Public Const CuChr = 1 '1byte 符号無
Public Const CuInt = 2 '2Byte 符号付
Public Const CuWrd = 4 '2Byte 符号無
Public Const CuLng = 8 '4Byte 符号付
    
```

◆Form1

USB のオープン、CUnet 初期化等は Form_Load() で行っています。
Command1_Click() でデータを読み込みグラフに反映、Command2_Click() で Text1 の内容を CSV 保存します。

```
Private Sub Command1_Click()

    dc = cunet_in(2032, CuWrd)          ' CUnet グローバルエリア データ総数
    Label2.Caption = "data count " + CStr(dc)
    MSChart1.RowCount = 0              ' Clear
    MSChart1.RowCount = dc
    MSChart1.TitleText = CStr(Date)

    Text1.Text = ""
    s = Int(dc / 10)
    With MSChart1.Plot.Axis(VtChAxisIdX).CategoryScale
        .Auto = False                  ' 自動設定を解除
        .DivisionsPerLabel = s         ' ラベル間隔
        .DivisionsPerTick = s         ' 補助メモリ表示単位
    End With

    read_data dc                        ' データ取得 & グラフ描画

    usb_close

End Sub

Private Sub read_data(dc)
Dim ar(0 To 59) As Long

    s = "時間, 温度 1, 湿度 1, 温度 2, 湿度 2"
    rc = 1
    For ar_top = 1 To dc Step 15
        res = cunet_req_pnt(4, ar_top, ar(0))    ' MPC にデータを要求: 要求 SA, P() 先頭, 格納配列
        If res <> 0 Then
            Label2.Caption = "DATA Receive fail"
            Exit Sub
        End If

        For i = 0 To 14                      ' ar() には 15 点分のデータが入っている
            ii = i * 4
            TM = Format(Hex(ar(ii)), "00:00:00")    ' X 軸
            TEMP1 = (Int(ar(ii + 1) / &H10000)) / 10    ' Y 軸 上位 2byte
            HUM1 = (ar(ii + 1) And 65535) / 10    ' Y 軸 下位 2byte
            TEMP2 = (Int(ar(ii + 2) / &H10000)) / 10    ' Z 軸 上位 2byte
            HUM2 = (ar(ii + 2) And 65535) / 10    ' Z 軸 下位 2byte

            If TM <> "00:00:00" Then
                s = s + Chr(13) + Chr(10) + TM + _
                    ", " + Format(TEMP1, "00.0") + ", " + Format(HUM1, "00.0") + _
                    ", " + Format(TEMP2, "00.0") + ", " + Format(HUM2, "00.0")
            End If

            If rc = 1 Then                      ' 最初の時間から保存ファイル名を作成
                Text2.Text = Format(Date, "YYMMDD-") + Format(Hex(ar(ii)), "000000.csv")
            End If

            If (rc <= dc) Then
                With MSChart1                  ' グラフに入れる
                    .Row = rc
                    .RowLabel = TM
                    .Column = 1
                    .Data = TEMP1
                    .Column = 2
                    .Data = HUM1
                    .Column = 3
                    .Data = TEMP2
                    .Column = 4
                    .Data = HUM2
                End With
            End If
            rc = rc + 1
        Next i
    End For
End Sub
```

```

Next ar_top
Text1.Text = s
Text1.SelStart = Len(s)

End Sub

Private Sub Command2_Click()
    On Error GoTo err_handler
    Open Text2.Text For Output As #1
    ' debug では C:\Program Files\Microsoft Visual Studio\VB98 に保存される
    Print #1, Text1.Text
    Close #1
    Exit Sub
err_handler:
    MsgBox Err.Description, vbCritical
End Sub

Private Sub Form_Load()
    If usb_open <> 1 Then ' USB-CUnet OPEN
        Label1.Caption = "USB OPEN ERROR"
        Exit Sub
    End If

    Label1.Caption = "fw_ver:" + CStr(fw_ver) + " dll_ver:" + CStr(dll_ver)
    init_cunet 255, 0, 0 ' USB-CUnet Reset
    Sleep 500
    init_cunet 0, 4, 8 ' USB-CUnet initialize SA, OWN, EN
    Sleep 500
    Label2.Caption = ""
    Text1.Text = ""
    Text1.FontName = "MS ゴシック"
    Text1.FontSize = 9
    Text2.Text = ""

    With MSChart1
        .chartType = VtChChartType2dLine ' グラフ種類
        .chartType = VtChChartType3dLine
        .RandomFill = False
        .RowCount = 0 ' Clear
        .ColumnCount = 4
        .Column = 1
        .ColumnLabel = "温度 1"
        .Column = 2
        .ColumnLabel = "湿度 1"
        .Column = 3
        .ColumnLabel = "温度 2"
        .Column = 4
        .ColumnLabel = "湿度 2"
        .ShowLegend = True ' 凡例表示
    End With
    With MSChart1.Plot.Axis(VtChAxisIdX).Labels(1).VtFont
        .Effect = VtFontEffectUnderline ' 文字飾り
        .Name = "MS ゴシック" ' フォント名
        .Size = 12 ' フォントサイズ
        .Style = VtFontStyleItalic ' スタイル
        .VtColor.Automatic = False ' 色の自動設定を解除
        .VtColor.Set 0, 255, 255 ' 色を設定
    End With
    With MSChart1.Plot.Axis(VtChAxisIdY).ValueScale
        .Auto = False ' 自動設定を解除
        .Maximum = 100 ' 最大値
        .Minimum = -10 ' 最小値
        .MajorDivision = 11 ' 目盛線数
        .MinorDivision = 2 ' 補助目盛線数
    End With
End Sub

```

■考察

隙間風をシャットアウトできればエコで快適... しっかり目張りしましょうか。